NAG Library Routine Document

F07MJF (DSYTRI)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F07MJF (DSYTRI) computes the inverse of a real symmetric indefinite matrix A, where A has been factorized by F07MDF (DSYTRF).

2 Specification

```
SUBROUTINE F07MJF(UPLO, N, A, LDA, IPIV, WORK, INFO)
INTEGER            N, LDA, IPIV(*), INFO
double precision   A(LDA,*), WORK(N)
CHARACTER*1        UPLO
```

The routine may be called by its LAPACK name ***dsytri***.

3 Description

F07MJF (DSYTRI) is used to compute the inverse of a real symmetric indefinite matrix A, the routine must be preceded by a call to F07MDF (DSYTRF), which computes the Bunch–Kaufman factorization of A.

If UPLO = 'U', $A = PUDU^{\mathrm{T}}P^{\mathrm{T}}$ and A^{-1} is computed by solving $U^{\mathrm{T}}P^{\mathrm{T}}XPU = D^{-1}$ for X.

If UPLO = 'L', $A = PLDL^{\mathrm{T}}P^{\mathrm{T}}$ and A^{-1} is computed by solving $L^{\mathrm{T}}P^{\mathrm{T}}XPL = D^{-1}$ for X.

4 References

Du Croz J J and Higham N J (1992) Stability of methods for matrix inversion *IMA J. Numer. Anal.* **12** 1–19

5 Parameters

1: UPLO – CHARACTER*1 *Input*

On entry: indicates how A has been factorized.

UPLO = 'U'
$A = PUDU^{\mathrm{T}}P^{\mathrm{T}}$, where U is upper triangular.

UPLO = 'L'
$A = PLDL^{\mathrm{T}}P^{\mathrm{T}}$, where L is lower triangular.

Constraint: UPLO = 'U' or 'L'.

2: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: N ≥ 0.

3: A(LDA,$*$) – ***double precision*** array *Input/Output*

Note: the second dimension of the array A must be at least max(1, N).

On entry: details of the factorization of A, as returned by F07MDF (DSYTRF).

On exit: the factorization is overwritten by the n by n symmetric matrix A^{-1}.

If UPLO = 'U', the upper triangle of A^{-1} is stored in the upper triangular part of the array.

If UPLO = 'L', the lower triangle of A^{-1} is stored in the lower triangular part of the array.

4: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F07MJF (DSYTRI) is called.

Constraint: LDA $\geq \max(1, \mathrm{N})$.

5: IPIV$(*)$ – INTEGER array *Input*

Note: the dimension of the array IPIV must be at least $\max(1, \mathrm{N})$.

On entry: details of the interchanges and the block structure of D, as returned by F07MDF (DSYTRF).

6: WORK(N) – ***double precision*** array *Workspace*

7: INFO – INTEGER *Output*

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO $= -i$, the ith parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

If INFO $= i$, $d(i,i)$ is exactly zero; D is singular and the inverse of A cannot be computed.

7 Accuracy

The computed inverse X satisfies a bound of the form

if UPLO = 'U', $\left|DU^{\mathrm{T}}P^{\mathrm{T}}XPU - I\right| \leq c(n)\epsilon\left(|D|\left|U^{\mathrm{T}}\right|P^{\mathrm{T}}|X|P|U| + |D|\left|D^{-1}\right|\right)$;

if UPLO = 'L', $\left|DL^{\mathrm{T}}P^{\mathrm{T}}XPL - I\right| \leq c(n)\epsilon\left(|D|\left|L^{\mathrm{T}}\right|P^{\mathrm{T}}|X|P|L| + |D|\left|D^{-1}\right|\right)$,

$c(n)$ is a modest linear function of n, and ϵ is the ***machine precision***.

8 Further Comments

The total number of floating-point operations is approximately $\frac{2}{3}n^3$.

The complex analogues of this routine are F07MWF (ZHETRI) for Hermitian matrices and F07NWF (ZSYTRI) for symmetric matrices.

9 Example

This example computes the inverse of the matrix A, where

$$A = \begin{pmatrix} 2.07 & 3.87 & 4.20 & -1.15 \\ 3.87 & -0.21 & 1.87 & 0.63 \\ 4.20 & 1.87 & 1.15 & 2.06 \\ -1.15 & 0.63 & 2.06 & -1.81 \end{pmatrix}.$$

Here A is symmetric indefinite and must first be factorized by F07MDF (DSYTRF).

9.1 Program Text

```
*     F07MJF Example Program Text
*     Mark 15 Release. NAG Copyright 1991.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX, LDA, LWORK
      PARAMETER        (NMAX=8,LDA=NMAX,LWORK=64*NMAX)
*     .. Local Scalars ..
      INTEGER          I, IFAIL, INFO, J, N
      CHARACTER        UPLO
*     .. Local Arrays ..
      DOUBLE PRECISION A(LDA,NMAX), WORK(LWORK)
      INTEGER          IPIV(NMAX)
*     .. External Subroutines ..
      EXTERNAL         DSYTRF, DSYTRI, X04CAF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F07MJF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      IF (N.LE.NMAX) THEN
*
*        Read A from data file
*
         READ (NIN,*) UPLO
         IF (UPLO.EQ.'U') THEN
            READ (NIN,*) ((A(I,J),J=I,N),I=1,N)
         ELSE IF (UPLO.EQ.'L') THEN
            READ (NIN,*) ((A(I,J),J=1,I),I=1,N)
         END IF
*
*        Factorize A
*
         CALL DSYTRF(UPLO,N,A,LDA,IPIV,WORK,LWORK,INFO)
*
         WRITE (NOUT,*)
         IF (INFO.EQ.0) THEN
*
*           Compute inverse of A
*
            CALL DSYTRI(UPLO,N,A,LDA,IPIV,WORK,INFO)
*
*           Print inverse
*
            IFAIL = 0
            CALL X04CAF(UPLO,'Nonunit',N,N,A,LDA,'Inverse',IFAIL)
         ELSE
            WRITE (NOUT,*) 'The factor D is singular'
         END IF
      END IF
*
      END
```

9.2 Program Data

```
F07MJF Example Program Data
  4                           :Value of N
  'L'                         :Value of UPLO
  2.07
  3.87  -0.21
  4.20   1.87   1.15
 -1.15   0.63   2.06  -1.81   :End of matrix A
```

9.3 Program Results

```
 F07MJF Example Program Results

 Inverse
             1          2          3          4
 1      0.7485
 2      0.5221    -0.1605
 3     -1.0058    -0.3131     1.3501
 4     -1.4386    -0.7440     2.0667     2.4547
```